# $Id: PKGBUILD 86622 2013-03-20 17:33:00Z eric $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>
# Contributor: Justin Davis <jrcd83@gmail.com>

pkgname='perl-test-requires'
pkgver='0.06'
pkgrel='1'
pkgdesc="Checks to see if the module can be loaded"
arch=('any')
license=('PerlArtistic' 'GPL')
options=('!emptydirs')
depends=('perl')
url='http://search.cpan.org/dist/Test-Requires'
source=("http://search.cpan.org/CPAN/authors/id/T/TO/TOKUHIROM/Test-Requires-$pkgver.tar.gz")
md5sums=('6ce0da3cceadb6420d4c3c5bb69f64db')

build() {
  cd "${srcdir}/Test-Requires-$pkgver"
  export PERL_MM_USE_DEFAULT=1 PERL5LIB=""                 \
    PERL_AUTOINSTALL=--skipdeps                            \
    PERL_MM_OPT="INSTALLDIRS=vendor"     \
    PERL_MB_OPT="--installdirs vendor" \
    MODULEBUILDRC=/dev/null
  perl Makefile.PL
  make
}

check() {
  cd "${srcdir}/Test-Requires-$pkgver"
  make test
}

package() {
  cd "${srcdir}/Test-Requires-$pkgver"
  make DESTDIR="$pkgdir" install
}
